

The_Red_Code

a *Mathematica* Notebook to study Particle Trajectories in Accelerator Magnets (*Approximate*)

Alessandro G. Ruggiero

Brookhaven National Laboratory

June 4, 2004

INITIALIZE

```
Off[General::"spell"]
Off[General::spell1]
Off[Part::"partw"]
Off[Plot::"plnr"]

$MachinePrecision
$MaxPrecision = $MinPrecision = 16;

15.9546
```

```

Clear[mDrift, mQuadF, mQuadD, mEdge, mSector, mQuad, mDipole, mTX]
Clear[x, h, k, ε, δ]
Clear[sequence, sequenceT, symmetry, nofelements]
Clear[xQF, kF, xQD, kD, xB, xD1, xD2, hB, angle]
Clear[kineticenergy, restenergy, gamma, beta, momentum, rigidity]
Clear[numberofperiods, dipolelength, quadlength,
driftlength, bendingangle, dipolefield, curvature]
Clear[periodlength, circumference, phaseadvance, kparameter, gradient]
Clear[namel, indx, length, curv, grad, angl]
Clear[s, t, find, fCurvature, fGradient, mElement, mTrans, mTransport]
Clear[ch, fpsih, cv, fpsiv, fbetaH, fbetaV,
falphaH, falphaV, fgammaH, fgammaV, fetaH, fetpH, falphaP]
Clear[mTransEta, mTransEtaP, mTransFP, mTransFPP,
mTransbtH, mTransbtV, mTransPsiH, mTransPsiV]
Clear[mTransAlphaH, mTransAlphaV, mTransPathL]
Clear[ψH, ψV, vH, vV, QH, QV, αP, γT]
Clear[eps, chromDH, chromDV, chromIH, chromIT]
Clear[plth, pltv, pltE]
Clear[ht, cl, mtblN, mR, R]
Clear[mTableH, Δp]
Clear[pltstl, plt, pltA]

```

INGREDIENTS

```

mDrift[x_, 0, 0, 0, 0] := {{1, x, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0},
{0, 0, 1, x, 0, 0}, {0, 0, 0, 1, 0, 0}, {0, 0, 0, 0, 1, 0}, {0, 0, 0, 0, 0, 1}}
mQuadF[x_, 0, k_, 0, δ_] :=
{{Cos[x Sqrt[k / (1 + δ)]], Sin[x Sqrt[k / (1 + δ)]] / Sqrt[k / (1 + δ)], 0, 0, 0, 0},
{- Sqrt[k / (1 + δ)] Sin[x Sqrt[k / (1 + δ)]], Cos[x Sqrt[k / (1 + δ)]], 0, 0, 0, 0},
{0, 0, Cosh[x Sqrt[k / (1 + δ)]], Sinh[x Sqrt[k / (1 + δ)]] / (Sqrt[k / (1 + δ)]), 0, 0},
{0, 0, Sqrt[k / (1 + δ)] Sinh[x Sqrt[k / (1 + δ)]], Cosh[x Sqrt[k / (1 + δ)]], 0, 0},
{0, 0, 0, 0, 1, 0}, {0, 0, 0, 0, 0, 1}}
mQuadD[x_, 0, k_, 0, δ_] := {{Cosh[x Sqrt[k / (1 + δ)]],
Sinh[x Sqrt[k / (1 + δ)]] / Sqrt[k / (1 + δ)], 0, 0, 0, 0},
{Sqrt[k / (1 + δ)] Sinh[x Sqrt[k / (1 + δ)]], Cosh[x Sqrt[k / (1 + δ)]], 0, 0, 0, 0},
{0, 0, Cos[x Sqrt[k / (1 + δ)]], Sin[x Sqrt[k / (1 + δ)]] / (Sqrt[k / (1 + δ)]), 0, 0},
{0, 0, -Sqrt[k / (1 + δ)] Sin[x Sqrt[k / (1 + δ)]], Cos[x Sqrt[k / (1 + δ)]], 0, 0},
{0, 0, 0, 0, 1, 0}, {0, 0, 0, 0, 0, 1}}
mEdge[0, h_, 0, ε_, δ_] := {{1, 0, 0, 0, 0, 0}, {h Tan[ε] / (1 + δ), 1, 0, 0, 0, 0},
{0, 0, 1, 0, 0, 0}, {0, 0, -h Tan[ε] / (1 + δ), 1, 0, 0},
{0, 0, 0, 0, 1, 0}, {0, 0, 0, 0, 0, 1}}
mSector[x_, h_, 0, 0, δ_] := {{Cos[x h / Sqrt[1 + δ]],
Sin[x h / Sqrt[1 + δ]] / (h / Sqrt[1 + δ]), 0, 0, 0, (1 - Cos[x h / Sqrt[1 + δ]]) / h},
{-h / Sqrt[1 + δ] Sin[x h / Sqrt[1 + δ]], Cos[x h / Sqrt[1 + δ]], 0, 0, 0,
Sin[x h / Sqrt[1 + δ]] / Sqrt[1 + δ]}, {0, 0, 1, x, 0, 0}, {0, 0, 0, 1, 0, 0},
{Sin[x h / Sqrt[1 + δ]] Sqrt[1 + δ], (1 + δ) (1 - Cos[x h / Sqrt[1 + δ]]) / h, 0,
0, 1, (x - Sin[x h / Sqrt[1 + δ]] / (h / Sqrt[1 + δ]))}, {0, 0, 0, 0, 0, 1}}
mQuad[x_, 0, k_, 0, δ_] := If[k ≥ 0, mQuadF[x, 0, k, 0, δ], mQuadD[x, 0, -k, 0, δ]]
mDipole[x_, h_, 0, ε_, δ_] :=
mEdge[0, h, 0, ε, δ] . mSector[x, h, 0, 0, δ] . mEdge[0, h, 0, ε, δ]

```

```

Array[mTX, 3];
mTX[1][x_, h_, k_, e_, δ_] := mDrift[x, 0, 0, 0, 0];
mTX[2][x_, h_, k_, e_, δ_] := mDipole[x, h, 0, e, δ];
mTX[3][x_, h_, k_, e_, δ_] := mQuad[x, 0, k, 0, δ];

MatrixForm[mDrift[x, 0, 0, 0, 0]];
MatrixForm[mEdge[0, h, 0, e, δ]];
MatrixForm[mQuad[x, 0, k, 0, δ]];
MatrixForm[mSector[x, h, 0, 0, δ]];

```

$$\begin{pmatrix}
\cos\left[x\sqrt{\frac{k}{1+\delta}}\right] & \frac{\sin\left[x\sqrt{\frac{k}{1+\delta}}\right]}{\sqrt{\frac{k}{1+\delta}}} & 0 & 0 & 0 & 0 \\
-\sqrt{\frac{k}{1+\delta}}\sin\left[x\sqrt{\frac{k}{1+\delta}}\right] & \cos\left[x\sqrt{\frac{k}{1+\delta}}\right] & 0 & 0 & 0 & 0 \\
0 & 0 & \cosh\left[x\sqrt{\frac{k}{1+\delta}}\right] & \frac{\sinh\left[x\sqrt{\frac{k}{1+\delta}}\right]}{\sqrt{\frac{k}{1+\delta}}} & 0 & 0 \\
0 & 0 & \sqrt{\frac{k}{1+\delta}}\sinh\left[x\sqrt{\frac{k}{1+\delta}}\right] & \cosh\left[x\sqrt{\frac{k}{1+\delta}}\right] & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 0 & 1
\end{pmatrix}$$

INPUT

■ Element Sequence

```

sequence = {QuadF, Drift1, Dipole, Drift2, QuadD};
symmetry = True;
QuadF = {"QuadF", 3, xQF, 0, kF, 0};
Drift1 = {"Drift1", 1, xd1, 0, 0, 0};
Dipole = {"Dipole", 2, xB, hB, 0, angle};
Drift2 = {"Drift2", 1, xd2, 0, 0, 0};
QuadD = {"QuadD", 3, xQD, 0, kD, 0};

```

■ General Parameters

```

kineticenergy = 1.0; (* GeV *)
restenergy = 0.93826; (* GeV *)
gamma = 1 + kineticenergy / restenergy;
beta = Sqrt[1 - 1 / gamma^2];
momentum = beta gamma restenergy; (* GeV/c *)
rigidity = 3.3356 momentum; (* Tesla-m *)

numberofperiods = 27;
dipolelength = 2.5; (* m *)
quadlength = 0.5; (* m *)
driftlength = 1.0; (* m *)
bendingangle =  $\pi$  / numberofperiods; (* rad *)
dipolefield = bendingangle rigidity / dipolelength; (* Tesla *)
curvature = dipolefield / rigidity; (* m *)

periodlength = 2 (dipolelength + quadlength + 2 driftlength); (* m *)
circumference = numberofperiods periodlength; (* m *)
phaseadvance = 90; (* degrees *)
kparameter = 4 Sin[ $\pi$  phaseadvance / 360] / (periodlength quadlength); (* 1/m2 *)
gradient = kparameter rigidity; (* Tesla / m *)
(* *)

```

■ Substitution

```

xQF = quadlength / 2;
xQD = quadlength / 2;
kF = kparameter;
kD = -kparameter;
xd1 = driftlength;
xd2 = driftlength;
hB = curvature;
angle = bendingangle / 2;
xB = dipolelength;

```

SOLUTION

■ Progressive Transfer Matrices

```

sequenceT = If[symmetry == True, Join[sequence, Reverse[sequence]], sequence];
numberofelements = Length[sequenceT];

Array[indx, numberofelements]; Array[length, numberofelements];
Array[curv, numberofelements]; Array[grad, numberofelements];
Array[angl, numberofelements]; Array[namel, numberofelements];
Array[s, numberofelements + 1, 0];

```

```

Do[namel[i] = sequenceT[[i]][[1]], {i, numberofelements}];
Do[indx[i] = sequenceT[[i]][[2]], {i, numberofelements}];
Do[length[i] = sequenceT[[i]][[3]], {i, numberofelements}];
s[0] = 0; Do[s[i] = s[i - 1] + length[i], {i, numberofelements}];
Do[curv[i] = sequenceT[[i]][[4]], {i, numberofelements}];
Do[grad[i] = sequenceT[[i]][[5]], {i, numberofelements}];
Do[angl[i] = sequenceT[[i]][[6]], {i, numberofelements}];
periodlength = s[numberofelements];
find[t_] := Which[t ≤ s[1], 1, s[1] < t && t ≤ s[2], 2, s[2] < t && t ≤ s[3],
  3, s[3] < t && t ≤ s[4], 4, s[4] < t && t ≤ s[5], 5, s[5] < t && t ≤ s[6], 6,
  s[6] < t && t ≤ s[7], 7, s[7] < t && t ≤ s[8], 8, s[8] < t && t ≤ s[9], 9, s[9] < t, 10];
fCurvature[t_] := (j = find[t]; curv[j]);
fGradient[t_] := (j = find[t]; grad[j]);

Array[mElement, numberofelements];
Array[mTrans, numberofelements + 1, 0];
Do[mElement[i][x_, h_, k_, e_, δ_] := Evaluate[mTX[indx[i]][x, h, k, e, δ]],
  {i, numberofelements}];
mTrans[0][δ_] := IdentityMatrix[6];
Do[mTrans[i][δ_] :=
  Evaluate[mElement[i][length[i], curv[i], grad[i], angl[i], δ] . mTrans[i - 1][δ]], {i,
  numberofelements}];

mTransport[t_, δ_] :=
  (i = find[t]; x = t - s[i - 1];
  mElement[i][x, curv[i], grad[i], angl[i], δ] . mTrans[i - 1][δ])

```

■ Momentum Dependence of Initial Twiss Parameters

```

Clear[t, δ]
t = periodlength;
cH[δ_] :=
  (mTransport[periodlength, δ][[1, 1]] + mTransport[periodlength, δ][[2, 2]]) / 2;
fpsiH[δ_] := 0.5 ArcCos[cH[δ]] / π;
cV[δ_] :=
  (mTransport[periodlength, δ][[3, 3]] + mTransport[periodlength, δ][[4, 4]]) / 2;
fpsiV[δ_] := 0.5 ArcCos[cV[δ]] / π;
fbetaH[δ_] := Abs[mTransport[periodlength, δ][[1, 2]]] / Sqrt[1 - cH[δ]^2];
fbetaV[δ_] := Abs[mTransport[periodlength, δ][[3, 4]]] / Sqrt[1 - cV[δ]^2];
falphaH[δ_] :=
  (mTransport[periodlength, δ][[1, 1]] - mTransport[periodlength, δ][[2, 2]]) /
  (2 Sqrt[1 - cH[δ]^2]);
falphaV[δ_] := (mTransport[periodlength, δ][[2, 2]] -
  mTransport[periodlength, δ][[4, 4]]) / (2 Sqrt[1 - cV[δ]^2]);
fgammaH[δ_] := (1 + falphaH[δ]^2) / fbetaH[δ];
fgammaV[δ_] := (1 + falphaV[δ]^2) / fbetaV[δ];
fetaH[δ_] := (mTransport[periodlength, δ][[1, 2]] mTransport[periodlength, δ][[2, 6]] -
  mTransport[periodlength, δ][[1, 6]]
  mTransport[periodlength, δ][[2, 2]] - 1) / (2 (1 - cH[δ]));
fetaV[δ_] := (mTransport[periodlength, δ][[2, 1]] mTransport[periodlength, δ][[1, 6]] -
  mTransport[periodlength, δ][[2, 6]]
  mTransport[periodlength, δ][[1, 1]] - 1) / (2 (1 - cH[δ]));
falphaP[δ_] := (mTransport[periodlength, δ][[5, 1]] fetaH[δ] +
  mTransport[periodlength, δ][[5, 2]] fetaV[δ] +
  mTransport[periodlength, δ][[5, 6]]) / periodlength;

```

Local Lattice Parameters

```

Clear[t, δ]
mTransEta[t_, δ_] := mTransport[t, δ][[1, 1]] fetaH[δ] +
  mTransport[t, δ][[1, 2]] fetapH[δ] + mTransport[t, δ][[1, 6]];
mTransEtaP[t_, δ_] := mTransport[t, δ][[2, 1]] fetaH[δ] +
  mTransport[t, δ][[2, 2]] fetapH[δ] + mTransport[t, δ][[2, 6]];
mTransFP[t_, δ_] := δ mTransEta[t, δ];
mTransFPP[t_, δ_] := δ mTransEtaP[t, δ];
mTransbtH[t_, δ_] := fbetaH[δ] mTransport[t, δ][[1, 1]]^2 - 2 mTransport[t, δ][[1, 1]]
  mTransport[t, δ][[1, 2]] falphaH[δ] + fgammaH[δ] mTransport[t, δ][[1, 2]]^2;
mTransbtV[t_, δ_] := fbetaV[δ] mTransport[t, δ][[3, 3]]^2 - 2 mTransport[t, δ][[3, 3]]
  mTransport[t, δ][[3, 4]] falphaV[δ] + fgammaV[δ] mTransport[t, δ][[3, 4]]^2;
mTransPsiH[t_, δ_] := ArcTan[fbetaH[δ] mTransport[t, δ][[1, 1]] -
  falphaH[δ] mTransport[t, δ][[1, 2]], mTransport[t, δ][[1, 2]]];
mTransPsiV[t_, δ_] := ArcTan[fbetaV[δ] mTransport[t, δ][[3, 3]] -
  falphaV[δ] mTransport[t, δ][[3, 4]], mTransport[t, δ][[3, 4]]];
mTransAlphaH[t_, δ_] := (1 + 2 mTransport[t, δ][[1, 2]] mTransport[t, δ][[2, 1]])
  falphaH[δ] - mTransport[t, δ][[2, 1]] mTransport[t, δ][[1, 1]] fbetaH[δ] -
  mTransport[t, δ][[1, 2]] mTransport[t, δ][[2, 2]] fgammaH[δ];
mTransAlphaV[t_, δ_] := (1 + 2 mTransport[t, δ][[3, 4]] mTransport[t, δ][[4, 3]])
  falphaV[δ] - mTransport[t, δ][[4, 3]] mTransport[t, δ][[3, 3]] fbetaV[δ] -
  mTransport[t, δ][[3, 4]] mTransport[t, δ][[4, 4]] fgammaV[δ];

```

Evolution of Path Length and Chromaticity

```

eps = 0.0001;
chromDH = 0.5 (fpsiH[eps] - fpsiH[-eps]) / (eps fpsiH[0]);
chromDV = 0.5 (fpsiV[eps] - fpsiV[-eps]) / (eps fpsiV[0]);

chromIH = 0; chromIV = 0;
Do[chromIH += +curv[i] Tan[angl[i]] (mTransbtH[s[i-1], 0] + mTransbtH[s[i], 0]),
  {i, numberofelements}];
Do[chromIV += -curv[i] Tan[angl[i]] (mTransbtV[s[i-1], 0] + mTransbtV[s[i], 0]),
  {i, numberofelements}];
chromIH += +NIntegrate[fGradient[u] mTransbtH[u, 0], {u, 0, periodlength}];
chromIV += -NIntegrate[fGradient[u] mTransbtV[u, 0], {u, 0, periodlength}];
chromIH = -360 chromIH / (4 π fpsiH[0]);
chromIV = -360 chromIV / (4 π fpsiV[0]);

mTransPathL[δ_] := NIntegrate[
  Sqrt[(1 + fCurvature[u] mTransFP[u, δ])^2 + mTransFPP[u, δ]^2], {u, 0, periodlength}];

```

RESULTS

Kinematics

E_0 (GeV)	kin. Energy (GeV)	β	γ	p (GeV/c)	$B\rho$ (T-m)
0.93826	1.	0.875027	2.0658	1.69603	5.65728

■ Lattice Sequence

QuadF Drift1 Dipole Drift2 QuadD QuadD Drift2 Dipole Drift1 QuadF

Name	Type	x = Length(m)	h = Curvature(1/m)	k = B'/Bρ(1/m ²)	ε = Angle(rad)
QuadF	3.	0.25	0.	0.565685	0.
Drift1	1.	1.	0.	0.	0.
Dipole	2.	2.5	0.0465421	0.	0.0581776
Drift2	1.	1.	0.	0.	0.
QuadD	3.	0.25	0.	-0.565685	0.

Circumference(m)	Periodicity	Period Length(m)	Symmetry
270.	27	10.	True


■ Transfer Matrix (versus t and δ)

$$\begin{pmatrix} 0.0690362 & 16.6103 & 0. & 0. & 0. & 1.55584 \\ -0.0599165 & 0.0690362 & 0. & 0. & 0. & 0.100134 \\ 0. & 0. & 0.0220353 & 3.08564 & 0. & 0. \\ 0. & 0. & -0.323925 & 0.0220353 & 0. & 0. \\ 0.100134 & 1.55584 & 0. & 0. & 1. & 0.103409 \\ 0. & 0. & 0. & 0. & 0. & 1. \end{pmatrix}$$

■ Global Parameters (versus δ)





ψ_H (°)	ψ_V (°)	γ_H	γ_V	Q_H	Q_V
86.0414	88.7374	0.239004	0.246493	6.4531	6.6553
ξ_H -Diff.	ξ_V -Diff.	ξ_H -Int.	ξ_V -Int.	α_P	γ_T
-1.24117	-1.2222	-1.24649	-1.15779	0.0270754	6.07732

■ Plot of Lattice Functions (versus δ)



■ Table of Lattice Parameters ($\delta = 0$)

#	s (m)	γ_H	γ_V	β_H (m)	β_V (m)	η (m)
0	0	0	0.0127386	16.072	3.20527	1.64176
1	0.25	0.00241808	0.0127386	16.072	3.20527	1.64176
2	1.25	0.0139417	0.0546428	11.8893	4.60595	1.40681
3	3.75	0.0673267	0.110602	4.82964	11.4401	0.966022
4	4.75	0.107302	0.122583	3.35192	15.4549	0.847553
5	5.	0.119502	0.125097	3.21718	16.0103	0.832788
6	5.25	0.131702	0.127612	3.35192	15.4547	0.847553
7	6.25	0.171677	0.139593	4.82964	11.4391	0.966022
8	8.75	0.225062	0.195566	11.8893	4.60423	1.40681
9	9.75	0.236586	0.237488	16.072	3.20374	1.64176
10	10.	0.239004	0.250233	16.6501	3.0849	1.67122

■ Plots of general Parameters versus δ 

■ Beam Envelopes versus δ and Period Length

